

## Part IV. Linear Differential Equations

### Section 4. Matrix Methods

In this section, you will learn enough about vectors and matrices in *Mathematica* to use them to obtain vector solutions to linear systems of the form

$$x' = a x + b y$$

$$y' = c x + d y$$

where  $x$  and  $y$  are functions of  $t$  and  $a, b, c, d$  are constants. Three dimensional systems will also be considered. Vectors and matrices arise naturally by replacing scalar equations with one vector equation  $v' = Av$ , where  $v$  is a vector valued function having the unknown functions as its components and  $A$  is a square matrix.

**Warning:** It is assumed that the reader knows what eigenvectors and eigenvalues are and understands the role they play in formulating solutions to systems of differential equations. (See Simmons/Krantz, *Anatomy of an Application*, Chapter 10.) Our principal goal is to demonstrate how to handle vectors, matrices, eigenvectors, etc. so that *Mathematica* can be applied successfully to the analysis of such systems.

#### Matrices and Vectors

A vector in *Mathematica* is represented as a list.

```
In[4]:= u = {1, 2, 3}
        v = {a, b, c}
```

```
Out[4]= {1, 2, 3}
```

```
Out[5]= {a, b, c}
```

The dot product is obtained as  $u.v$  (put a period between  $u$  and  $v$ ).

```
In[3]:= u.v
```

```
Out[3]= a + 2 b + 3 c
```

A matrix is a list of lists. Each list is a row.

```
In[1]:= A = {{1, 2, 3}, {4, 5, 6}, {7, 8, 9}}
```

```
Out[1]= {{1, 2, 3}, {4, 5, 6}, {7, 8, 9}}
```

The function `MatrixForm` will display  $A$  as a matrix.

```
In[2]:= MatrixForm[A]
```

```
Out[2]//MatrixForm=
```

$$\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{pmatrix}$$

The `MatrixForm` function displays a one-dimensional list (i.e. a vector) as a column. The next entry shows an alternate way to apply the `MatrixForm` function.

---

In[6]:= **u//MatrixForm**

Out[6]//MatrixForm=

$$\begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix}$$

The matrix A and the vector v are multiplied in this order using A.v. *Mathematica* interprets v as column vector and outputs the product as a vector (i.e. a list).

In[6]:= **A.v**

Out[6]= {a + 2 b + 3 c, 4 a + 5 b + 6 c, 7 a + 8 b + 9 c}

In[7]:= **v.A//MatrixForm**

Out[7]//MatrixForm=

$$\begin{pmatrix} a + 4 b + 7 c \\ 2 a + 5 b + 8 c \\ 3 a + 6 b + 9 c \end{pmatrix}$$

Matrix A and vector v can also be multiplied in the opposite order using v.A. Now *Mathematica* interprets v as a row and still outputs the product as a vector.

In[8]:= **v.A**

Out[8]= {a + 4 b + 7 c, 2 a + 5 b + 8 c, 3 a + 6 b + 9 c}

The MatrixForm function displays this product as a column also.

In[9]:= **v.A//MatrixForm**

Out[9]//MatrixForm=

$$\begin{pmatrix} a + 4 b + 7 c \\ 2 a + 5 b + 8 c \\ 3 a + 6 b + 9 c \end{pmatrix}$$

The Table function can be used to make matrices.

In[10]:= **B = Table[ 1/(m+n^2), {m,1,3}, {n,0,2} ]**

Out[10]= {{1,  $\frac{1}{2}$ ,  $\frac{1}{5}$ }, { $\frac{1}{2}$ ,  $\frac{1}{3}$ ,  $\frac{1}{6}$ }, { $\frac{1}{3}$ ,  $\frac{1}{4}$ ,  $\frac{1}{7}$ }}

The determinant of B is obtained with the entry

Det[B]

---

In[11]:= **MatrixForm[B]**  
**Det[B]**

Out[11]//MatrixForm=

$$\begin{pmatrix} 1 & \frac{1}{2} & \frac{1}{5} \\ \frac{1}{2} & \frac{1}{3} & \frac{1}{6} \\ \frac{1}{3} & \frac{1}{4} & \frac{1}{7} \end{pmatrix}$$

Out[12]=  $\frac{1}{1260}$

Since B's determinant is not zero, it has an inverse. It can be found using

Inverse[B]

In[13]:= **Binverse = Inverse[B]**  
**B.Binverse**

Out[13]=  $\left\{ \left\{ \frac{15}{2}, -27, 21 \right\}, \{-20, 96, -84\}, \left\{ \frac{35}{2}, -105, 105 \right\} \right\}$

Out[14]=  $\left\{ \{1, 0, 0\}, \{0, 1, 0\}, \{0, 0, 1\} \right\}$

Be careful because the entry 1/B or B<sup>(-1)</sup> will output the matrix whose entries are the reciprocals of the entries in B.

In[15]:= **1/B**

Out[15]=  $\left\{ \{1, 2, 5\}, \{2, 3, 6\}, \{3, 4, 7\} \right\}$

To find B's characteristic polynomial, as a function of *t*, enter

CharacteristicPolynomial[B,t]

In[29]:= **CharacteristicPolynomial[B,t]**

Out[29]=  $\frac{1}{1260} - \frac{139 t}{840} + \frac{31 t^2}{21} - t^3$

The Euclidean length of a vector *v* can be found using Norm[v].

In[31]:= **v**  
**Norm[v]**

Out[31]=  $\{a, b, c\}$

Out[32]=  $\sqrt{\text{Abs}[a]^2 + \text{Abs}[b]^2 + \text{Abs}[c]^2}$

The absolute value function is applied to the components *a*, *b*, *c* because they might be complex. If *a*, *b*, *c* are known to be real numbers, then the Euclidean norm of *v* can also be found as follows.

In[33]:= **Sqrt[v.v]**

Out[33]=  $\sqrt{a^2 + b^2 + c^2}$

The entry Norm[v,∞] calculates what is called the "infinity" or "max" norm of *v*. This is the maximum of the absolute values of the entries.

---

In[36]:= **u**  
**Norm[u, ∞]**

Out[36]= {1, 2, 3}

Out[37]= 3

Note. The infinity symbol can be entered using the symbol palette. Typing the word "infinity" will not work.

The entry B[[i,j]] outputs the i,j entry of matrix B. The entry v[[k]] is the kth component of the vector v.

In[43]:= **{B[[2,2]], v[[2]]}**

Out[43]=  $\left\{\frac{1}{3}, b\right\}$

The jth column of matrix B is obtained using Take[B,All,{j,j}]. The entry Take[B,{i,i}] outputs its ith row. They appear as lists.

In[51]:= **Take[B,All,{3,3}]**  
**Take[B,{1,1}]**

Out[51]=  $\left\{\left\{\frac{1}{5}\right\}, \left\{\frac{1}{6}\right\}, \left\{\frac{1}{7}\right\}\right\}$

Out[52]=  $\left\{\left\{1, \frac{1}{2}, \frac{1}{5}\right\}\right\}$

*Augmenting and stacking (not needed later, but kind of neat anyway)*

If you would like to stack the matrix B with the row vector {x,y,z}, on the bottom, do it like this.

In[75]:= **Bstacked = Insert[B,{x,y,z},4]**

Out[75]=  $\left\{\left\{1, \frac{1}{2}, \frac{1}{5}\right\}, \left\{\frac{1}{2}, \frac{1}{3}, \frac{1}{6}\right\}, \left\{\frac{1}{3}, \frac{1}{4}, \frac{1}{7}\right\}, \{x, y, z\}\right\}$

In[76]:= **MatrixForm[Bstacked]**

Out[76]//MatrixForm=

$$\begin{pmatrix} 1 & \frac{1}{2} & \frac{1}{5} \\ \frac{1}{2} & \frac{1}{3} & \frac{1}{6} \\ \frac{1}{3} & \frac{1}{4} & \frac{1}{7} \\ x & y & z \end{pmatrix}$$

If you would like to then put {a,b,c,d} into the right column of Bstacked, first transpose Bstacked.

In[79]:= **Transpose[Bstacked]**

Out[79]=  $\left\{\left\{1, \frac{1}{2}, \frac{1}{3}, x\right\}, \left\{\frac{1}{2}, \frac{1}{3}, \frac{1}{4}, y\right\}, \left\{\frac{1}{5}, \frac{1}{6}, \frac{1}{7}, z\right\}\right\}$

Then insert {a,b,c,d} along the bottom.

In[80]:= **Insert[%,{a,b,c,d},4]**

Out[80]=  $\left\{\left\{1, \frac{1}{2}, \frac{1}{3}, x\right\}, \left\{\frac{1}{2}, \frac{1}{3}, \frac{1}{4}, y\right\}, \left\{\frac{1}{5}, \frac{1}{6}, \frac{1}{7}, z\right\}, \{a, b, c, d\}\right\}$

And then transpose again.

---

```
In[81]:= Bstackedaugmented = Transpose[%]
```

```
Out[81]= {{1, 1/2, 1/5, a}, {1/2, 1/3, 1/6, b}, {1/3, 1/4, 1/7, c}, {x, y, z, d}}
```

```
In[82]:= %//MatrixForm
```

```
Out[82]//MatrixForm=
```

$$\begin{pmatrix} 1 & \frac{1}{2} & \frac{1}{5} & a \\ \frac{1}{2} & \frac{1}{3} & \frac{1}{6} & b \\ \frac{1}{3} & \frac{1}{4} & \frac{1}{7} & c \\ x & y & z & d \end{pmatrix}$$

Finally, if you need to calculate the determinant of the 3 x 3 submatrix of Bstackedaugmented that is obtained by deleting its first row and first column, do it like this.

```
In[92]:= Bstackedaugmented[[{2,3,4},{2,3,4}]];
```

```
%//MatrixForm
```

```
%%//Det
```

```
Out[93]//MatrixForm=
```

$$\begin{pmatrix} \frac{1}{3} & \frac{1}{6} & b \\ \frac{1}{4} & \frac{1}{7} & c \\ y & z & d \end{pmatrix}$$

```
Out[94]=  $\frac{d}{168} - \frac{by}{7} + \frac{cy}{6} + \frac{bz}{4} - \frac{cz}{3}$ 
```

## Eigenvectors and Eigenvalues

The matrix A, defined below, is square. It has eigenvectors and eigenvalues. Let's find them.

```
In[96]:= A = {{3,2,1},{0,-1,-1},{2,1,1}};
```

```
%//MatrixForm
```

```
Out[97]//MatrixForm=
```

$$\begin{pmatrix} 3 & 2 & 1 \\ 0 & -1 & -1 \\ 2 & 1 & 1 \end{pmatrix}$$

First find A's characteristic polynomial.

```
In[3]:= cp = CharacteristicPolynomial[A,t]
```

```
Out[3]=  $-2 + 2t + 3t^2 - t^3$ 
```

Then obtain the eigenvalues as its roots. Call the output lambda.

```
In[12]:= Solve[cp==0,t]
```

```
Out[12]= {{t -> -1}, {t -> 2 -  $\sqrt{2}$ }, {t -> 2 +  $\sqrt{2}$ }}
```

And use it to make a list of eigenvalues called lambda.

---

```
In[13]:= lambda = Table[t/.%[[j]],{j,1,3}]
```

```
Out[13]= {-1, 2 -  $\sqrt{2}$ , 2 +  $\sqrt{2}$ }
```

Now use the **NullSpace** function to obtain bases for the three eigenspaces. Recall that the kth eigenspace is the null space of the matrix

$$\mathbf{A} - \text{lambda}[[k]]*\text{Identity}$$

where "Identity" is the 3 x 3 identity matrix. This is obtained in *Mathematica* with the entry

**IdentityMatrix[3].**

```
In[35]:= v = Table[ NullSpace[ A - lambda[[k]]*IdentityMatrix[3] ],
                 {k,1,3} ]
```

```
Out[35]= {{{-1, 2, 0}}, {{- $\frac{3-2\sqrt{2}}{-3+\sqrt{2}}$ ,  $\frac{1}{-3+\sqrt{2}}$ , 1}}, {{- $\frac{-3-2\sqrt{2}}{3+\sqrt{2}}$ , - $\frac{1}{3+\sqrt{2}}$ , 1}}}}
```

The output for **NullSpace** is a list containing a basis for the null space, in list form. If further computations are required it would probably be wise to convert the eigenvalues and eigenvectors to floating point form.

```
In[39]:= N[lambda, 3]
         N[v, 3]
```

```
Out[39]= {-1.00, 0.586, 3.41}
```

```
Out[40]= {{{-1.00, 2.00, 0}}, {{0.1082, -0.631, 1.00}}, {{1.320, -0.2265, 1.00}}}}
```

### Mathematica's direct method

*Mathematica* has functions that will output eigenvectors and eigenvalues directly. To obtain A's eigenvalues apply the **Eigenvalues** function.

```
In[41]:= Eigenvalues[A]
```

```
Out[41]= {2 +  $\sqrt{2}$ , -1, 2 -  $\sqrt{2}$ }
```

The output is a list of A's eigenvalues. The function called **Eigenvectors** will output a list with each eigenvector appearing in list form. Observe that the eigenvectors appear in simplified form and in the order corresponding to the appearance of the output for **Eigenvalue**.

---

In[42]:= **Eigenvectors[A]**

Out[42]=  $\left\{ \left\{ -\frac{1}{7} + \frac{3}{7} (2 + \sqrt{2}), -\frac{5}{7} + \frac{1}{7} (2 + \sqrt{2}), 1 \right\}, \right.$   
 $\left. \{-1, 2, 0\}, \left\{ -\frac{1}{7} + \frac{3}{7} (2 - \sqrt{2}), -\frac{5}{7} + \frac{1}{7} (2 - \sqrt{2}), 1 \right\} \right\}$

### Eigenvector solutions to $v' = Av$ , the 2 x 2 case

Let's begin by solving a simple 2 x 2 system

$$x' = x + 2y$$

$$y' = x - y$$

using eigenvectors. This is equivalent to  $v' = Av$  where  $v = \langle x, y \rangle$  and

$$A = \begin{pmatrix} 1 & 2 \\ 1 & -1 \end{pmatrix}.$$

Start with the eigenvectors and eigenvalues of A.

In[108]:= **A = {{1,2},{1,-1}};**  
**%//MatrixForm**  
**lambda = Eigenvalues[A]**  
**v = Eigenvectors[A]**

Out[109]//MatrixForm=

$$\begin{pmatrix} 1 & 2 \\ 1 & -1 \end{pmatrix}$$

Out[110]=  $\{-\sqrt{3}, \sqrt{3}\}$

Out[111]=  $\left\{ \{1 - \sqrt{3}, 1\}, \{1 + \sqrt{3}, 1\} \right\}$

Let  $v_1$  and  $v_2$  be the eigenvector solutions. They are straight line trajectories. Recall that  $v_k$  is defined using the formula

$$v_k = e^{\lambda_k t} E_k$$

where  $E_k$  is an eigenvector for the eigenvalue  $\lambda_k$ .

```
In[51]:= v1 = Exp[lambda[[1]]*t]*v[[1]]
         v2 = Exp[lambda[[2]]*t]*v[[2]]
```

```
Out[51]= {(1 - sqrt(3)) e^{-sqrt(3) t}, e^{-sqrt(3) t}}
```

```
Out[52]= {(1 + sqrt(3)) e^{sqrt(3) t}, e^{sqrt(3) t}}
```

Now that we have  $v_1$  and  $v_2$ , the easiest way to obtain solutions to the IVP

$$v' = Av, v(0) = v_0$$

is to use the formula

$$v(t) = X(t) X(0)^{-1} v(0)$$

where  $X(t)$  is the fundamental matrix having  $v_1$  and  $v_2$  as its columns.


```
In[65]:= X = Transpose[{v1, v2}]
```

```
Out[65]= {{(1 - sqrt(3)) e^{-sqrt(3) t}, (1 + sqrt(3)) e^{sqrt(3) t}}, {e^{-sqrt(3) t}, e^{sqrt(3) t}}}
```

For example, the following entries define the vector  $v$  as the solution to  $v' = Av$ ,  $v(0) = \langle 1, 1 \rangle$ , then show the trajectory and the initial point. The formula for the solution is suppressed because it is a mess.

```
In[89]:= v = X.Inverse[X/.t->0].{1,1};
```

```
In[86]:= Pt = ListPlot[{{1,1}}, PlotStyle->PointSize[0.03]];
         Traj = ParametricPlot[v, {t,-1,1}, PlotStyle->RGBColor[0,1,0] ];
         SolnTraj = Show[ Pt, Traj, PlotRange->{{-2,4},{-2,4}}, AspectRatio->1/1 ]
```



Here is the solution vector, simplified.


```
In[90]:= Simplify[v];
          %//MatrixForm
```

```
Out[91]//MatrixForm=
```

$$\begin{pmatrix} \frac{1}{2} e^{-\sqrt{3} t} (1 - \sqrt{3} + (1 + \sqrt{3}) e^{2\sqrt{3} t}) \\ \frac{1}{2} e^{-\sqrt{3} t} (1 + e^{2\sqrt{3} t}) \end{pmatrix}$$

And here is a picture of the solution trajectory and two eigenvector trajectories, one blue, one red.

```
In[93]:= Show[ ParametricPlot[ v1, {t,-5,1}, PlotStyle->RGBColor[0,0,1] ],
               ParametricPlot[ v2, {t,-5,1}, PlotStyle->RGBColor[1,0,0] ],
               SolnTraj, PlotRange->{{-2,4},{-2,4}}, AspectRatio->1/1]
```



### The matrix exponential

The function called **MatrixExp**, when applied to square matrix A as

**MatrixExp[t\*A]**

outputs the matrix  $X(t)X(0)^{-1}$  that was found above using the eigenvector solutions. We compute it below, applied the 2 x 2 matrix A defined above.

```
In[112]:= Mexp = MatrixExp[t*A];
          Simplify[%]//MatrixForm
```

```
Out[113]//MatrixForm=
```

$$\begin{pmatrix} \frac{1}{6} e^{-\sqrt{3} t} (3 - \sqrt{3} + (3 + \sqrt{3}) e^{2\sqrt{3} t}) & \frac{e^{-\sqrt{3} t} (-1 + e^{2\sqrt{3} t})}{\sqrt{3}} \\ \frac{e^{-\sqrt{3} t} (-1 + e^{2\sqrt{3} t})}{2\sqrt{3}} & \frac{1}{6} e^{-\sqrt{3} t} (3 + \sqrt{3} - (-3 + \sqrt{3}) e^{2\sqrt{3} t}) \end{pmatrix}$$

Compare the matrix Mexp to  $X(t)X(0)^{-1}$ , simplified and in matrix form.

In[105]:= **X.Inverse[X/.t->0]//Simplify//MatrixForm**

Out[105]//MatrixForm=

$$\begin{pmatrix} \frac{1}{6} e^{-\sqrt{3} t} (3 - \sqrt{3} + (3 + \sqrt{3}) e^{2\sqrt{3} t}) & \frac{e^{-\sqrt{3} t} (-1 + e^{2\sqrt{3} t})}{\sqrt{3}} \\ \frac{e^{-\sqrt{3} t} (-1 + e^{2\sqrt{3} t})}{2\sqrt{3}} & \frac{1}{6} e^{-\sqrt{3} t} (3 + \sqrt{3} - (-3 + \sqrt{3}) e^{2\sqrt{3} t}) \end{pmatrix}$$

### A 3 dimensional example, approximate solutions

We will now compute three linearly independent eigenvector solutions to  $v' = A v$  for the 3 x 3 matrix A defined several pages ago and redefined below.

In[106]:= **A = {{3,2,1},{0,-1,-1},{2,1,1}};**  
**%//MatrixForm**

Out[107]//MatrixForm=

$$\begin{pmatrix} 3 & 2 & 1 \\ 0 & -1 & -1 \\ 2 & 1 & 1 \end{pmatrix}$$

Because this is a small system, exact formulas can be obtained. However we will make our calculations in floating point form. We do this for three reasons.

1. Floating point calculations are easier to make, and are more relevant for real world applications.
2. It will give us some experience with error analysis.
3. If exact solution formulas are required, it is easier to get them directly from the matrix exponential or by simply applying DSolve.

Here, once more, are A's eigenvalues and eigenvectors, but in approximate decimal form.

In[108]:= **lambda = N[Eigenvalues[A]]**  
**EV = N[Eigenvectors[A]]**

Out[108]= {3.41421, -1., 0.585786}

Out[109]= {{1.32038, -0.226541, 1.}, {-1., 2., 0.}, {0.108194, -0.630602, 1.}}

The following Table constructs three eigenvector solutions.

In[110]:= **v = Table[ Exp[lambda[[k]]\*t]\*EV[[k]], {k,1,3}]**

Out[110]= {{1.32038 e<sup>3.41421 t</sup>, -0.226541 e<sup>3.41421 t</sup>, 1. e<sup>3.41421 t</sup>}, {-1. e<sup>-1. t</sup>, 2. e<sup>-1. t</sup>, 0. e<sup>-1. t</sup>}, {0.108194 e<sup>0.585786 t</sup>, -0.630602 e<sup>0.585786 t</sup>, 1. e<sup>0.585786 t</sup>}}

The fundamental matrix with these vectors in its columns is called X.

```
In[111]:= X = Transpose[v];
          %//MatrixForm
```

```
Out[112]//MatrixForm=
```

$$\begin{pmatrix} 1.32038 e^{3.41421 t} & -1. e^{-1. t} & 0.108194 e^{0.585786 t} \\ -0.226541 e^{3.41421 t} & 2. e^{-1. t} & -0.630602 e^{0.585786 t} \\ 1. e^{3.41421 t} & 0. e^{-1. t} & 1. e^{0.585786 t} \end{pmatrix}$$

And here is the solution to  $v' = Av$  satisfying  $v(0) = \langle 2, 4, 1 \rangle$ .

```
In[116]:= v = X.Inverse[X/.t->0].{2, 4, 1};
          Simplify[%]//MatrixForm
```

```
Out[117]//MatrixForm=
```

$$\begin{pmatrix} e^{-1. t} (-1.71429 - 0.21367 e^{1.58579 t} + 3.92796 e^{4.41421 t}) \\ e^{-1. t} (3.42857 + 1.24536 e^{1.58579 t} - 0.673931 e^{4.41421 t}) \\ 0. e^{-1. t} - 1.97487 e^{0.585786 t} + 2.97487 e^{3.41421 t} \end{pmatrix}$$

The next entry checks the initial value.

```
In[118]:= v/.t->0
```

```
Out[118]= {2., 4., 1.}
```

*Error analysis over the interval  $t = 0$  to  $t = 8$*

The vector function  $v$  does not satisfy the differential equation exactly. Let's see how large the error is over the time interval from 0 to 8. The error can be analyzed as follows. First calculate the "error vector"  $v' - Av$ .

```
In[128]:= D[v,t] - A.v;
          ErrorVector = Simplify[%];
          %//MatrixForm
```

```
Out[130]//MatrixForm=
```

$$\begin{pmatrix} 4.44089 \times 10^{-16} e^{-1. t} - 2.22045 \times 10^{-16} e^{0.585786 t} + 0. e^{3.41421 t} \\ -4.44089 \times 10^{-16} e^{-1. t} + 0. e^{0.585786 t} - 8.88178 \times 10^{-16} e^{3.41421 t} \\ 4.44089 \times 10^{-16} e^{-1. t} + 0. e^{0.585786 t} + 1.77636 \times 10^{-15} e^{3.41421 t} \end{pmatrix}$$

Then define Error as a function whose value at  $t$  is the largest of the absolute errors in the three components of ErrorVector. This requires the "infinity" norm.


```
In[131]:= Error = Norm[ErrorVector,∞]
```

```
Out[131]= Max[Abs[-4.44089 × 10-16 e-1. t + 0. e0.585786 t - 8.88178 × 10-16 e3.41421 t],
             Abs[4.44089 × 10-16 e-1. t - 2.22045 × 10-16 e0.585786 t + 0. e3.41421 t],
             Abs[4.44089 × 10-16 e-1. t + 0. e0.585786 t + 1.77636 × 10-15 e3.41421 t]]
```

Now graph Error over the interval from 0 to 8.

---

```
In[164]:= Plot[Error, {t,0,8}, PlotRange->{0,0.002}]
```



This looks very good. Although the error clearly grows exponentially, it is less than 0.0015 at  $t = 8$  (years, maybe?). We say years because at  $t = 8$  the approximate solution vector is over  $3.6 \times 10^{12}$  units long. (The length calculation uses the Euclidean norm.)

```
In[166]:= Norm[v/.t->8]
```

```
Out[166]= 3.62102 × 1012
```

Thus the maximum relative error over this interval could be as small as 4 parts in  $10^{16}$ .